Option Explicit
Const odstr_Zac = "<", odstr_Kon = ">", najdi_Nbsp = "&nbsp", nahrad_Nbsp = " "
Function Vycisti_http(Zdroj As String) As String
'funkce odstraní sekvence začínající < a končící > a nahradí je mezerou. _
    Nahradí "&nbsp" mezerou. Na závěr odstraní nadbytečné mezery.
Dim i As Integer, j As Integer, k As Integer, Pom As String, _
        OdstrString As String
'Kdyby se něco nedařilo, ponech řetěz beze změny
Vycisti_http = Zdroj
'Nejprve odstraň Nbsp
Vycisti_http = Replace(Vycisti_http, najdi_Nbsp, nahrad_Nbsp) 'nahradí _
                                            všechny výskyty &nbsp mezerou
'Abychom mohli utéct z nějakého problému, zapamatujeme si _
    rozpracovaný nový řetězec
Pom = Vycisti_http
'Nyní v cyklu hledej začátek odstraňované sekvence
j = InStr(1, Pom, odstr_Zac)
While j > 0
    k = InStr(1, Pom, odstr_Kon)   'najdeme konec nejlevější _
                            sekvence typu <Strong>, což je ">"
'musíme se ujistit, že j<k. V opačném případě je ve zdroji chyba, _
                                            kterou neumíme opravit.
    If j < k Then
       OdstrString = Mid(Pom, j, k - j + 1)
       Vycisti_http = Replace(Pom, OdstrString, " ")
       Pom = Vycisti_http
    Else    'chybný řetězec - vyznačíme uživateli problém
        Vycisti_http = "CHYBA! " & Vycisti_http
        Exit Function
    End If
'Odstranili jsme všechny výskyty první nalezené sekvence. _
    Zkusíme, zda existuje ještě jiná
    j = InStr(1, Pom, odstr_Zac)    'když nic nenajde, j=0 a konec cyklu
Wend
Vycisti_http = Application.WorksheetFunction.Trim( _
        Application.WorksheetFunction.Clean(Vycisti_http))
End Function
